Problem:
Score: 500 points

Problem Statement

We have caught N sardines. The deliciousness and fragrantness of the i-th sardine is A_i and B_i, respectively.

We will choose one or more of these sardines and put them into a cooler. However, two sardines on bad terms cannot be chosen at the same time.

The i-th and j-th sardines (i \neq j) are on bad terms if and only if A_i \cdot A_j + B_i \cdot B_j = 0.

In how many ways can we choose the set of sardines to put into the cooler? Since the count can be enormous, print it modulo 1000000007.

Constraints

All values in input are integers.

1 \leq N \leq 2 \times 10^5

-10^{18} \leq A_i, B_i \leq 10^{18}

Input

Input is given from Standard Input in the following format:

N
A_1 B_1
:
A_N B_N

Output

Print the count modulo 1000000007.

Sample Input 1

3
1 2
-1 1
2 -1

Sample Output 1

5

There are five ways to choose the set of sardines, as follows:

The 1-st

The 1-st and 2-nd

The 2-nd

The 2-nd and 3-rd

The 3-rd

Sample Input 2

10
3 2
3 2
-1 1
2 -1
-3 -9
-8 12
7 7
8 1
8 2
8 4

Sample Output 2

479

Solution in OCaml (4.10.0) (judge verdict: Accepted)

let rec gcd n m =
  if m = 0 then n else gcd m (n mod m)

let m = 1000000007
let ( +^ ) x y = (x + y) mod m
let ( -^ ) x y = x +^ (m - y)
let ( *^ ) x y = (x * y) mod m

let rec power ( * ) e m n =
  if n <= 0 then e
  else power ( * ) (if n land 1 = 0 then e else m * e) (m * m) (n lsr 1)

type quotient = Nan | Zero | Inf | Quotient of int * int

let quotient_of_intpair = function
  | (0, 0) -> Nan
  | (0, _) -> Zero
  | (_, 0) -> Inf
  | (a, b) ->
      let d = gcd (abs a) (abs b) in
      if b < 0
      then Quotient (~- a / d, ~-b / d)
      else Quotient (a / d, b / d)

let inv = function
  | Zero -> Inf
  | Inf -> Zero
  | Quotient (a, b) -> Quotient (b * compare 0 a, abs a)

module QuotientMap = Map.Make (struct
  type t = quotient
  let compare = compare
end)

let () = Scanf.scanf "%d\n" @@ fun n ->
  let abs = List.init n @@ fun _ -> Scanf.scanf "%d %d\n" @@ fun a b -> quotient_of_intpair (a, b) in
  let zeros, m =
    List.fold_left (fun (zeros, m) -> function
      | Nan -> (zeros + 1, m)
      | q ->
          zeros,
          QuotientMap.update q (fun o -> Some (1 + Option.value ~default:0 o)) @@
          QuotientMap.update (inv q) (fun o -> Some (Option.value ~default:0 o)) m) (0, QuotientMap.empty) abs in
  Printf.printf "%d\n" @@
  ( +^ ) (zeros -^ 1) @@
  QuotientMap.fold (fun q n acc ->
    match q with
    | Zero -> acc
    | Quotient (a, _) when a < 0 -> acc
    | _ ->
        let n' = QuotientMap.find (inv q) m in
        acc *^ (power ( *^ ) 1 2 n +^ power ( *^ ) 1 2 n' -^ 1)) m 1